
=====================Repeatable_Randomness=====================================

1) Pseudorandom sequence generating circuits repeat themselves over time.
2) The number of flips flops used defines the length of the randomness.
3) Pseudorandomness can modulate a signal to look like noise (spread spectrum modulation).
4) Spread spectrum modulated signal can be demodulated back into signal using the same 
   pseudorandom sequence generating circuit.    

*                   pseudorandom sequence generator
*           __
*    xnor  / //_________________________________________
*  _____/\/  ||                                         |
* |     \/\  ||_________________________________        |
* |V0      \_\\                                 |  ETC  |
* |   ___ V1   ___ V2   ___  V3  ___   V4 ___   |  ___  |
* |__|D Q|____|D Q|____|D Q|____|D Q|____|D Q|__|_|D Q|_|
*    |   |    |   |    |   |    |   |    |   |    |   |
*  _\|C  |  _\|C  |  _\|C  |  _\|C  |  _\|C  |  _\|C  |
* | /|___| | /|___| | /|___| | /|___| | /|___| | /|___|
* |________|________|________|________|________|
* STAGES           LENGTH      FEEDBACK
* 3                     7               2,3
* 4                     15              3,4
* 5                     31              3,5
* 6                     63              5,6
* 7                     127             6,7 

===================MacSpiceCode====================================================
Repeatable_Digital_Randomness 
*                   pseudorandom sequence generator
*           __
*    xnor  / //_________________________________________
*  _____/\/  ||                                         |
* |     \/\  ||_________________________________        |
* |V0      \_\\                                 |  ETC  |
* |   ___ V1   ___ V2   ___  V3  ___   V4 ___   |  ___  |
* |__|D Q|____|D Q|____|D Q|____|D Q|____|D Q|__|_|D Q|_|
*    |   |    |   |    |   |    |   |    |   |    |   |
*  _\|C  |  _\|C  |  _\|C  |  _\|C  |  _\|C  |  _\|C  |
* | /|___| | /|___| | /|___| | /|___| | /|___| | /|___|
* |________|________|________|________|________|
* STAGES           LENGTH      FEEDBACK
* 3                     7               2,3
* 4                     15              3,4
* 5                     31              3,5



* 6                     63              5,6
* 7                     127             6,7

*=========Create_Signal==================
*.OPTIONS GMIN=1p      METHOD=trap   ABSTOL=1m     TEMP=27   srcsteps = 1  gminsteps = 1 
*.OPTIONS RELTOL=.001  ABSTOL=1p     VNTOL=1u      ITL4=500  ITL1=400 
*V_PULSE# NODE_P NODE_N DC     VALUE  PULSE( VINIT  VPULSE TDELAY TRISE  TFALL  PWIDTH PERIOD )
VCLK      CLK    0      DC     0      PULSE( 0      1      1n     1n     1n     1m     2m )

XDFF1     V0     CLK    V1     D_FF
XDFF2     V1     CLK    V2     D_FF
XDFF3     V2     CLK    V3     D_FF
XDFF4     V3     CLK    V4     D_FF
XDFF5     V4     CLK    V5     D_FF
XDFF6     V5     CLK    V6     D_FF
BA        A      0      V =    v(V3)
BB        B      0      V =    v(V2)

BXNOR     V0     0      V =    1-u( v(A)*(.5-v(B)) + v(B)*(.5-v(A))-.1)

.control
set       pensize = 2
tran      1u     50m    0        1u
plot      clk    v1-1.2   
.endc

*=========D_FF============================================
*            
* This subcircuit uses a break before make to converge !!!!
*                                             
*  ___   |\ VC0/    VC1 |\  VC2 /    VC3  |\      ___
* |D  |__|1\__/  _______|2\ ___/  ________|4\____|Q  |
* |___|  | /  S1   _|_  | /  S2   _|_     | /    |___|
*        |/        ___  |/        ___     |/
*                C1 |           C2 |
*                  _|_            _|_
*                  ///            ///                  |\
*  ___   |\             VBF ____               DPE   __|B\__\S2
* |CLK|__|3\________________|   \ PE                |  | /  /
* |___|  | / |           VLP| PE \___/\  /\  /\_____|  |/
*        |/  |/\  /\  /\____|    /     \/  \/   |   |
*               \/  \/   |  |___/      RLP2    _|_  |  |\
*               RLP1    _|_                    ___  |__|M\__\S1
*                       ___                CLP2 |      | /  /
*                   CLP1 |                     _|_     |/
*                       _|_                    ///
*                       ///
.SUBCKT   D_FF    D      CLK         Q      
 B3       VBF     0      V =    u( v(CLK )-.5 )
 RLP      VBF     VLP    1k
 CLP      VLP     0      5n  
 BAND     VPE     0      V =    u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.5)
 RLP2     VPE     VLP2   .3k
 CLP2     VLP2    0      2n 
 BBRK     BRK     0      V =     1-u(v(VLP2 )-.2)
 BMAK     MAK     0      V =     u(v(VLP2 )-.9)
 B1       VC0     0      V =     u( v(D  )-.5 )
 S1       VC0     VC1    MAK     0        SW 
 C1       VC1     0      10n
 R1       VC1     0      10Meg
 B2       VC2     0      V =     u( v(VC1)-.5 )
 S2       VC2     VC3    BRK     0        SW 
 C2       VC3     0      1n
 R2       VC3     0      10Meg
 B4       Q       0      V =     u(v(VC3)-.5)
.ENDS     D_FF

.MODEL    SW      SW(    VT=.5 VH=.1  RON=10 ROFF=100MEG)

.end

